

Postestimation commands

The following postestimation commands are available after `proportion`:

Command	Description
<code>contrast</code>	contrasts and ANOVA-style joint tests of parameters
<code>estat vce</code>	variance–covariance matrix of the estimators (VCE)
<code>estat (svy)</code>	postestimation statistics for survey data
<code>estimates</code>	cataloging estimation results
<code>etable</code>	table of estimation results
<code>lincom</code>	point estimates, standard errors, testing, and inference for linear combinations of parameters
<code>marginsplot</code>	graph the results from <code>proportion</code>
<code>nlcom</code>	point estimates, standard errors, testing, and inference for nonlinear combinations of parameters
<code>pwcompare</code>	pairwise comparisons of parameters
<code>test</code>	Wald tests of simple and composite linear hypotheses
<code>testnl</code>	Wald tests of nonlinear hypotheses

Remarks and examples

► Example 1

In [example 2](#) of [\[R\] proportion](#), we computed the proportions of cars with different repair records for each group, foreign or domestic. We use `test` to test whether the proportion of cars with repair record equal to 4 is the same for domestic and foreign cars.

```
. use https://www.stata-press.com/data/r19/auto
(1978 automobile data)

. proportion rep78, over(foreign)
(output omitted)

. test 4.rep78@0.foreign=4.rep78@1.foreign
( 1)  4.rep78@0bn.foreign - 4.rep78@1.foreign = 0
      F(  1,    68) =    3.92
      Prob > F =    0.0518
```

There is not a significant difference between those proportions at the 5% level.



► Example 2

Continuing with `auto.dta` from [example 1](#), we generate a new variable, `highprice`, that indicates if the price is larger than \$5,000 and then use `proportion` to see the proportion of cars with high price among domestic and foreign cars separately.

```
. generate highprice = price>5000
. proportion highprice, over(foreign)
Proportion estimation                                Number of obs = 74
```

	Proportion	Std. err.	Logit [95% conf. interval]	
highprice@foreign				
0 Domestic	.5576923	.0688744	.4195373	.6874611
0 Foreign	.3636364	.1025593	.191094	.5802222
1 Domestic	.4423077	.0688744	.3125389	.5804627
1 Foreign	.6363636	.1025593	.4197778	.808906

We will compute the odds ratio of having a high price in group `Foreign` to having a high price in group `Domestic`. Usually, odds ratios are computed by using the `logistic` command, but here we will perform the computation by using `nlcom` after `proportion`.

```
. nlcom OR: (_b[1.highprice@1.foreign]/_b[0.highprice@1.foreign])/(_b[1.highpri
> ce@0.foreign]/_b[0.highprice@0.foreign])
      OR: (_b[1.highprice@1.foreign]/_b[0.highprice@1.foreign])/(_b[1.high
> price@0.foreign]/_b[0.highprice@0.foreign])
```

Proportion	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
OR	2.206522	1.155825	1.91	0.056	-.0588533	4.471897

This is the same odds ratio that we would obtain from

```
. logistic highprice foreign
```

The odds ratio is slightly larger than 2, which means that the odds of having a high price among foreign cars are more than twice that of having a high price among domestic cars.

◀

Also see

[\[R\] proportion](#) — Estimate proportions

[\[U\] 20 Estimation and postestimation commands](#)

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

